# $Id$
# Maintainer:  Bartłomiej Piotrowski <bpiotrowski@archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>

pkgname=expect
pkgver=5.45
pkgrel=5
pkgdesc='A tool for automating interactive applications'
arch=(i686 x86_64)
url='http://www.nist.gov/el/msid/expect.cfm'
license=(custom)
depends=(tcl)
source=(http://downloads.sourceforge.net/project/expect/Expect/$pkgver/expect${pkgver}.tar.gz)
md5sums=('44e1a4f4c877e9ddc5a542dfa7ecc92b')

build() {
  cd expect${pkgver}
  ./configure --prefix=/usr --mandir=/usr/share/man 
  make
}

check() {
  cd expect${pkgver}
  make test
}

package() {
  cd expect${pkgver}
  make DESTDIR="$pkgdir" install
  install -Dm644 README "$pkgdir/usr/share/licenses/$pkgname/README"
}
